A Program for Test of Common Factors of Two Polynomials

Söderström, Torsten

1973

Document Version:
Publisher's PDF, also known as Version of record

Link to publication

Citation for published version (APA):
Söderström, T. (1973). *A Program for Test of Common Factors of Two Polynomials.* (Technical Reports TFRT-7050). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:
1

LUND UNIVERSITY

PO Box 117
221 00 Lund
+46 46-222 00 00

A PROGRAM FOR TEST OF COMMON FACTORS OF TWO POLYNOMIALS

T. SÖDERSTRÖM

Report 7313 (C) July 1973
Lund Institute of Technology
Division of Automatic Control

A PROGRAM FOR TEST OF COMMON FACTORS OF TWO POLYNOMIALS.

T. Söderström

ABSTRACT.

In this report a program for test of common factors of two polynomials is given.

TABLE OF CONTENTS

I. DESCRIPTION OF THE PROGRAMS.

The method used for the test is described in Söderström (1973 b). Some approximate methods are considered in Burström (1973). These approximate methods, however, are not to be preferred. The programs SYMIN, DECOM, SOLVB and FLET are given in the program library of the Division of Automatic Control. The program FACT is described in Söderström (1973 a).

1. TCFAC. This is a test program which uses the subroutines SIMUL, PRBSTA, PRB, NODI, LS and LSQ of the program library of the Division of Automatic Control to generate a LS estimate as input data for CFAC.

2. CFAC. It is assumed that the two polynomials

$$A(z) = z^{NA} + a_1 z^{NA-1} + \dots + a_{NA}$$

$$B(z) = b_0 z^{NB} + b_1 z^{NB-1} + \dots + b_{NB}$$

are given. The vector

$$T = [a_1 \dots a_{NA} \; b_1 \dots b_{NB}]$$

is assumed to have a known covariance matrix P.

The coefficient b_0 is known. It is either 0 or 1.

It is tested if A(z) and B(z) have K common factors in a statistical sense. The test is performed as an optimization of a non quadratic loss function using the subroutine FLET. The resulting polynomials

$$Al(z) = z^{NA-K} + al_1 z^{NA-K-1} + \ldots + al_{NA-K}$$

$$Bl(z) = b_0 z^{NB-K} + bl_1 z^{NB-K-1} + \ldots + bl_{NB-K}$$

$$C(z) = z^K + c_1 z^{K-1} + \ldots + c_K$$

satisfy approximately

$$A(z) = Al(z) \cdot C(z)$$

$$B(z) = Bl(z) \cdot C(z)$$

The coefficients are stored in the vector TR as

$$TR = [al_1 \ldots al_{NA-K}\ bl_1 \ldots bl_{NB-K}\ c_1 \ldots c_K]$$

3. <u>VCFAC</u>. This routine computes the loss function for the minimization and its gradient.

II. REFERENCES.

Burström, E. (1973).
Förkortning av gemensamma faktorer i skattade överförings-funktioner. Master Thesis RE 124, Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 a).
A Program Package for GLS Identification of a Dynamic System. Report 7311(C), Division of Automatic Control, Lund Institute of Technology.

Söderström, T. (1973 b).
On the Simplification of Dynamic Models Obtained by Least Squares Identification (preliminary title). Forthcoming report, Division of Automatic Control, Lund Institute of Technology.

III. LISTS OF THE PROGRAMS

```
C       TEST OF ALTERNATIVE GLS
C
        DIMENSION U(1000),Y(1000),DAT(3000),AB(1000,9)
        DIMENSION TS(30),TM(30),TMR(30)
        COMMON / LSCOM/ V,S,P(50,50),C(50),Q(50)
C
    1   READ 100,NS,NM,M
  100   FORMAT(3I5)
        IF(M.EQ.0) STOP
        NS3=NS*3
        READ 101,(TS(I),I=1,NS3),AL
  101   FORMAT(8F10.5)
C
        PRINT 200
  200   FORMAT(1H1,10X,'SYSTEM PARAMETERS')
        PRINT 101,(TS(I),I=1,NS3),AL
        PRINT 100,NS,NM,M
        IE=3323
        CALL SIMUL(U,Y,TS,1.0,AL,M,NS,NS,NS,1,7,IE)
        DO 2 I=1,M
        DAT(3*I-1)=U(I)
    2   DAT(3*I)=Y(I)
        NM2=NM*2
        K1=MSLEFT(DUM)
        CALL LS(DAT,TM,AB,M,1,NM2,NM2,1000,9,1)
        CALL CFAC(TM,TMR,P,V,NM2,NM2,NM,0,50,IFA,1.0)
        K2=MSLEFT(DUM)
        K=K1-K2
        PRINT 300,K
  300   FORMAT(//10X,'TIME FOR IDENTIFICATION WAS',I5,' MS')
        GO TO 1
        END
```

```
      SUBROUTINE CFAC(T,TR,P,V,NA,NB,K,IB,IA,IFAIL,TSAMP)
C
C     GIVEN
C     TWO POLYNOMIALS
C     A(Z)=Z**NA+T(1)*Z**(NA-1)+...+T(NA)
C     B(Z)=B0**NB+T(NA+1)*Z**(NB-1)+...T(NA+NB)
C     AND THE COVARIANCE MATRIX P OF THE VECTOR T
C
C     IT IS TESTED IF A(Z) AND B(Z) HAVE K COMMON ZEROS
C     THREE NEW POLYNOMIALS ARE COMPUTED. THEY SATISFY APPROXIMATELY
C     A(Z) = A1(Z)*C(Z)    B(Z) = B1(Z)*C(Z)
C     A1(Z) = Z**(NA-K)+TR(1)*Z**(NA-K-1)+...+TR(NA-K)
C     B1(Z) = B0*Z**(NB-K)+TR(NA-K+1)*Z**(NB-K-1)+...+TR(NA-K+NB-K)
C     C(Z) = Z**K+TR(NA+NB-2*K+1)*Z**(K-1)+...+TR(NA+NB-K)
C
C     AUTHOR TORSTEN SöDERSTRöM 1973-04-01
C     REFERENCE REPORT 7313(C)
C
C     T-VECTOR OF ORDER (NA+NB)
C     TR-VECTOR OF  ORDER (NA+NB-K)
C     P MATRIX OF ORDER (NA+NB)*(NA+NB)
C     V-LOSS FUNCTION AND TEST QUANTITY
C       UNDER SUITABLE CONDITIONS V IS CHI SQUARE DISTRIBUTED
C       WITH K DEGREES OF FREEDOM
C     NA ORDER OF A(Z)   (MIN 1,MAX 10)
C     NB ORDER OF B(Z)   (MIN 1,MAX 10)
C     K  ORDER OF C(Z)   (MIN 1,MAX 10)
C     IB- IF IB=0  B0=0
C         IF IB=1  B0=1
C     IA DIMENSION PARAMETER
C     IFAIL - IF IFAIL=0 THE SUBROUTINE HAS PROCEEDED WITHOUT ERRORS
C             IF IFAIL=1 ERRORS HAVE OCCURED IN THE COMPUTATIONS
C     TSAMP - THE SAMPLING INTERVAL (MAY BE GIVEN AN ARBITRARY POSITIVE VALUE)
C
C     IF IB=0 K.LE.MIN(NA,NB-1)
C     IF IB=1 K.LE.MIN(NA,NB)
C
C     SUBROUTINE REQUIRED
C              SYMIN
C              DECOM
C              SOLVB
C              FLET
C              VCFAC
C              FACT
C
      DIMENSION T(1),TR(1),P(IA,IA)
      COMMON /CCFAC/ Q(20,20),TT(20),NA1,NB1,KK,BB0
      DIMENSION R(20,20),RB(20),H(210),G(20),W(80),EPS1(20)
      EXTERNAL VCFAC
C
      IFAIL=0
      PRINT 210
  210 FORMAT(///10X,'PRINTOUT FROM SUBROUTINE CFAC',10X,29(1H*)/10X,29(1
     *H*))
      PRINT 211
  211 FORMAT(/10X,'ORIGINAL A-POLYNOMIAL')
      PRINT 100,(T(I),I=1,NA)
      PRINT 212
  212 FORMAT(/10X,'ORIGINAL B-POLYNOMIAL')
      PRINT 100,(T(I+NA),I=1,NB)
      PRINT 213,IB
  213 FORMAT(/10X,'B(0)=',I1)
      PRINT 214,K
```

```
  214 FORMAT(/10X,'IT IS TESTED IF A AND B HAVE',I3,' COMMON FACTORS')
C
      B0=FLOAT(IB)
      KMAX=MIN0(NA,NB-1+IB)
      IF(K.GT.0.AND.K.LE.KMAX) GO TO 1
      PRINT 900,K
  900 FORMAT(/10X,'K=',I2,' IS NOT SUITABLE')
      IFAIL=1
      GO TO 99
    1 CONTINUE
      NA1=NA-K
      NB1=NB-K
      M=NA+NB
      M1=NA1+NB1
      MM=M1+K
      EPS=1.0E-05
      DO 2 I=1,M
    2 TT(I)=T(I)
      KK=K
      BB0=B0
      DO 3 I=1,MM
    3 EPS1(I)=1.0E-05
C
      DO 4 I=1,M
      DO 4 J=1,M
    4 Q(I,J)=P(I,J)
      CALL SYMIN(M,20,IFAIL,Q)
      IF(IFAIL.EQ.0) GO TO 10
      PRINT 901
  901 FORMAT(/10X,'THE MATRIX P IS SINGULAR')
      GO TO 99
C
C     COMPUTE INITIAL VALUES
C
   10 IF(M1.EQ.0) GO TO 20
      DO 11 I=1,M1
      RB(I)=0.
      DO 11 J=1,M1
   11 R(I,J)=0.
      IF(NA1.EQ.0) GO TO 13
      DO 12 I=1,NA1
      R(I,I)=-B0
      MJ=MIN0(M1-I,NB)
      DO 12 J=1,MJ
   12 R(I+J,I)=-T(NA+J)
   13 IF(NB1.EQ.0) GO TO 15
      DO 14 I=1,NB1
      R(I,NA1+I)=1.
      MJ=MIN0(M1-I,NA)
      DO 14 J=1,MJ
   14 R(I+J,NA1+I)=T(J)
   15 MJ=MIN0(M1,NA)
      DO 16 J=1,MJ
   16 RB(J)=-T(J)*B0
      MJ=MIN0(M1,NB)
      DO 17 J=1,MJ
   17 RB(J)=RB(J)+T(NA+J)
      CALL DECOM(R,M1,20,EPS,ISING)
      IF(ISING.EQ.0) GO TO 18
      PRINT 902
  902 FORMAT(/10X,'DECOMPOSITION IS IMPOSSIBLE'/
     *10X,'K MAY BE CHOSEN TOO SMALL')
      IFAIL=1
      GO TO 99
```

```
   18 CALL SOLVB(RB,TR,M1,1,20)
C
   20 CONTINUE
      DO 21 J=1,K
   21 RB(J)=0.
      MJ=MIN0(NA,K)
      DO 22 J=1,MJ
   22 RB(J)=T(J)
      MJ=MIN0(NA1,K)
      IF(MJ.EQ.0) GO TO 24
      DO 23 J=1,MJ
   23 RB(J)=RB(J)-TR(J)
   24 CONTINUE
      DO 25 I=1,K
      TR(M1+I)=RB(I)
      MJ=MIN0(I-1,NA1)
      IF(MJ.EQ.0) GO TO 25
      DO 26 J=1,MJ
   26 TR(M1+I)=TR(M1+I)-TR(J)*TR(M1+I-J)
   25 CONTINUE
C
C
      PRINT 200
  200 FORMAT(/10X,'INITIAL VALUES')
      PRINT 100,(TR(I),I=1,MM)
  100 FORMAT(10G12.5)
C
      HEPS=Q(1,1)
      MH=MM*(MM+1)/2
      DO 31 I=1,MH
   31 H(I)=0.
      J=-MM
      DO 32 I=1,MM
      J=J+MM+2-I
   32 H(J)=HEPS
C
C     MINIMIZATION
C
      IF(MM.EQ.1) GO TO 40
      DFN=HEPS
      MAXFN=200
      MODE=2
      IPRINT=MAXFN+1
      CALL FLET(VCFAC,MM,TR,V,G,H,W,DFN,EPS1,MODE,MAXFN,IPRINT,IEXIT)
      IF(IEXIT.NE.1) IFAIL=1
      IF(IEXIT.EQ.4) PRINT 206
  206 FORMAT(/5X,5(1H*),'WARNING  THE COVARIANCE MATRIX IS INDEFINITE')
      GO TO 50
C
   40 PP=Q(1,1)*T(1)+Q(1,2)*(T(1)+T(2))+Q(2,2)*T(2)
      PQ=Q(1,1)+2.*Q(1,2)+Q(2,2)
      TR(1)=PP/PQ
      PS=T(1)**2*Q(1,1)+T(2)**2*Q(2,2)+2.*T(1)*T(2)*Q(1,2)
      V=PS-PP**2/PQ
   50 CONTINUE
C
C     PRINT RESULT
C
      PRINT 201
  201 FORMAT(/10X,'PARAMETERS OF REDUCED MODEL')
      IF(NA1.EQ.0) GO TO 51
      PRINT 202
  202 FORMAT(/10X,'PARAMETERS OF THE NEW A POLYNOMIAL')
      PRINT 100,(TR(I),I=1,NA1)
```

```
   51 IF(NB1.EQ.0) GO TO 52
      PRINT 203
  203 FORMAT(/10X,'PARAMETERS OF THE NEW B POLYNOMIAL')
      PRINT 100,(TR(I+NA1),I=1,NB1)
   52 PRINT 204
  204 FORMAT(/10X,'PARAMETERS OF THE COMMON FACTOR')
      PRINT 100,(TR(I+M1),I=1,K)
      PRINT 205
  205 FORMAT(/10X,'LOSS FUNCTION AND TEST QUANTITY')
      PRINT 100,V
C
      IF(IB.EQ.1) GO TO 60
      CALL FACT(TR,NA1,NB1,K,TSAMP)
      GO TO 70
   60 IF(NA1.EQ.0) GO TO 62
      DO 61 I=1,NA1
   61 H(I)=TR(I)
   62 H(NA1+1)=1.
      IF(NB1.EQ.0) GO TO 64
      DO 63 I=1,NB1
   63 H(NA1+1+I)=TR(NA1+I)
   64 DO 65 I=1,K
   65 H(NA1+1+NB1+I)=TR(NA1+NB1+I)
      NB11=NB1+1
      CALL FACT(H,NA1,NB11,K,TSAMP)
C
C
   99 CONTINUE
   70 PRINT 220
  220 FORMAT(///10X,'END OF SUBROUTINE CFAC',10X,30(1H*)/10X,22(1H*))
C
C
      RETURN
      END
```

```
      SUBROUTINE VCFAC(TR,M1,V,G,FXX,IA,IB,IERR)
C
C     SUBROUTINE FOR CFAC
C     AUTHOR TORSTEN SÖDERSTRÖM 1973-04-01
C
C
      DIMENSION TR(1),G(1),FXX(IA,1)
      COMMON /CCFAC/ Q(20,20),T(20),NA1,NB1,K,B0
      DIMENSION S(20,20),TS(20),Y(20)
C
      NA=NA1+K
      NB=NB1+K
      M=NA+NB
      M1=NA1+NB1+K
      MA=NA1+NB1
C
      DO 1 I=1,M
      DO 1 J=1,M1
    1 S(I,J)=0.
      IF(NA1.EQ.0) GO TO 3
      DO 2 I=1,NA1
      S(I,I)=1.
      DO 2 J=1,K
    2 S(I+J,I)=TR(MA+J)
    3 IF(NB1.EQ.0) GO TO 5
      DO 4 I=1,NB1
      S(NA+I,NA1+I)=1.
      DO 4 J=1,K
    4 S(NA+I+J,NA1+I)=TR(MA+J)
    5 DO 9 I=1,K
      S(I,MA+I)=1.
      S(NA+I,MA+I)=B0
      IF(NA1.EQ.0) GO TO 7
      DO 6 J=1,NA1
    6 S(I+J,MA+I)=TR(J)
    7 IF(NB1.EQ.0) GO TO 9
      DO 8 J=1,NB1
    8 S(NA+I+J,MA+I)=TR(NA1+J)
    9 CONTINUE
C
      DO 11 I=1,M
   11 TS(I)=T(I)
      IF(NA1.EQ.0) GO TO 13
      DO 12 I=1,NA1
   12 TS(I)=TS(I)-TR(I)
   13 IF(NB1.EQ.0) GO TO 15
      DO 14 I=1,NB1
   14 TS(I+NA)=TS(I+NA)-TR(I+NA1)
   15 CONTINUE
      DO 16 I=1,M
      DO 16 J=1,K
   16 TS(I)=TS(I)-S(I,MA+J)*TR(MA+J)
      DO 17 I=1,M
      Y(I)=0.
      DO 17 J=1,M
   17 Y(I)=Y(I)+TS(J)*Q(J,I)
C
      V=0.
      DO 21 I=1,M
   21 V=V+Y(I)*TS(I)
      DO 22 I=1,M1
      G(I)=0.
      DO 22 J=1,M
```

```
   22 G(I)=G(I)-2.0*Y(J)*S(J,I)
C
      IF(V.LT.0) IERR=1
C
      RETURN
      END
```

IV. EXAMPLES

Example 1. The true system is $y(t)-1.0y(t-1)+0.25y(t-2) = 1.0u(t-1)-0.5u(t-1) + \lambda e(t)$ with $\lambda=3.0$ The number of data is 500.

```
         SYSTEM PARAMETERS
-1.00000     .25000   1.00000    -.50000     .00000     .00000    3.00000
  2    1   500

        RESULT FROM IDENTIFICATION, MODEL OF ORDER  2
          A( 1)=  -.99410              STAND DEV=   .43389-01
          A( 2)=    .25729             STAND DEV=   .42930-01
        B 1( 1)=   1.0285              STAND DEV=   .13807
        B 1( 2)=  -.46234              STAND DEV=   .14458

        VALUE OF LOSS FUNCTION     4670.7
        ESTIMATED STANDARD DEVIATION S=    3.0749

        SINGULAR VALUES
 155.59       56.346       21.087       22.272


        PRINTOUT FROM SUBROUTINE CFAC          ******************************
        ******************************

        ORIGINAL A-POLYNOMIAL
 -.99410       .25729

        ORIGINAL B-POLYNOMIAL
 1.0285       -.46234

        B(0)=0

        IT IS TESTED IF A AND B HAVE  1 COMMON FACTORS

        INITIAL VALUES
 -.54456       1.0285      -.44954

ENTRY TO FLET
   0    1
  .22534772+00
 -.54456070+00     .10284641+01    -.44954078+00
 -.11636394+02     .27534781-01    -.85678380+01
 15   16    1
  .16944606+00
 -.52655341+00     .10201566+01    -.46075859+00
 -.10073185-03    -.56706369-04    -.57697296-04
  .15078134+04     .54583628-01     .85403330+00     .13694675+03    -.31976499+00     .15165878+03

        PARAMETERS OF REDUCED MODEL

        PARAMETERS OF THE NEW A POLYNOMIAL
 -.52655

        PARAMETERS OF THE NEW B POLYNOMIAL
 1.0202

        PARAMETERS OF THE COMMON FACTOR
 -.46076

        LOSS FUNCTION AND TEST QUANTITY        (the same variable)
 .16945


        RESULT OF FACTORIZATION
        ************************


         POLES OF DISCRETE MODEL
  .52655       .00000


         POLES OF CONTINOUS MODEL
 -.64140       .00000


         ZEROS OF THE C-POLYNOMIAL
  .46076       .00000


         CONTINOUS EQUIVALENT
 -.77488       .00000


         STATIC GAIN
 2.1547


        END OF SUBROUTINE CFAC          ******************************
        **********************


        TIME FOR IDENTIFICATION WAS  520 MS
```

Example 2. The same system as in example 1 but $\lambda = 1.0$

```
         SYSTEM PARAMETERS
 -1.00000     .25000   1.00000   -.50000    .00000    .00000   1.00000
  2    1  500

         RESULT FROM IDENTIFICATION, MODEL OF ORDER  2
           A( 1)=  -.98541                 STAND DEV=   .42407-01
           A( 2)=   .23919                 STAND DEV=   .37813-01
         B 1( 1)=   1.0094                 STAND DEV=   .46027-01
         B 1( 2)=  -.47376                 STAND DEV=   .62468-01

         VALUE OF LOSS FUNCTION   519.08
         ESTIMATED STANDARD DEVIATION S=   1.0251

         SINGULAR VALUES
  59.090       30.424       13.580       22.288


         PRINTOUT FROM SUBROUTINE CFAC          ******************************
         ******************************

         ORIGINAL A-POLYNOMIAL
 -.98541       .23919

         ORIGINAL B-POLYNOMIAL
  1.0094     -.47376

         B(0)=0

         IT IS TESTED IF A AND B HAVE  1 COMMON FACTORS

         INITIAL VALUES
 -.51606       1.0094      -.46935

ENTRY TO FLET
   0    1
   .17833905-01
  -.51605982+00    .10093959+01   -.46935029+00
   .25830053+01    .66566654-02    .23433351+01
   8     9    1
   .15627421-01
  -.51638575+00    .10096027+01   -.47087545+00
   .39517879-04    .32205135-04   -.46546757-03
   .24437540+04    .23148080+00    .57216893+00    .10506644+04   -.23305535+00    .54586564+03

         PARAMETERS OF REDUCED MODEL

         PARAMETERS OF THE NEW A POLYNOMIAL
 -.51639

         PARAMETERS OF THE NEW B POLYNOMIAL
  1.0096

         PARAMETERS OF THE COMMON FACTOR
 -.47088

         LOSS FUNCTION AND TEST QUANTITY
  .15627-01


         RESULT OF FACTORIZATION
         ************************

         POLES OF DISCRETE MODEL
  .51639       .00000

         POLES OF CONTINOUS MODEL
 -.66090       .00000

         ZEROS OF THE C-POLYNOMIAL
  .47088       .00000

         CONTINOUS EQUIVALENT
 -.75316       .00000

         STATIC GAIN
  2.0876


         END OF SUBROUTINE CFAC          *******************************
         **********************

         TIME FOR IDENTIFICATION WAS  527 MS
```